
About this document 

This document describes how to run the IBM i Access Client Solutions Java–based data transfer program 
on the IBM i. 

Your IBM i server must be at OS/400 V5R4 or above to use the steps shown in 
this document. 

The IBM i Access Client Solutions (ACS) product is a new offering from IBM. It provides some of the 
features of the traditional IBM i Access for Windows (5250 emulation, data transfer) in a Java 
environment. Any client computer that has the Java 6 or Java 7 runtime environment can use the ACS 
programs. For example, the program can be installed on a Macintosh computer or on a PC running Linux. 

Because the product is entirely Java–based, it is also possible to run the Data Transfer features (file 
download / file upload) on the IBM i. It is important to understand that installing and running the IBM i 
Access Client Solutions product on the IBM i is not supported by IBM. 

Installing and using the ACS data transfer components on the IBM i may be useful for some scenarios 
where you need to extract data from the IBM i database into PC file formats, or where you need to upload 
data from PC file formats to the IBM i database. By moving the data transfer components to the IBM i, you 
can take advantage of the IBM i batch processes to run the data transfer programs. This may lessen the 
need for creating and maintaining cumbersome PC–based data transfer mechanisms, which traditionally 
use remote command features. 

The PC format files that are produced (file download) or consumed (file upload) are located in the IBM i 
IFS 

After installing and configuring the ACS components on the IBM i, you can use a workstation–based 
version of ACS to develop the data transfer applications. The outputs of the workstation development are 
PC files that describe the data transfers. After creating and saving the description files on the workstation, 
you copy the description files to the IBM i IFS. When you run the data transfer programs on the IBM i, you 
reference the description files. 

IMPORTANT 
USE THE STEPS DESCRIBED IN THIS DOCUMENT AT YOUR OWN RISK. 

IBM WILL NOT PROVIDE SUPPORT FOR THE STEPS DESCRIBED IN THIS DOCUMENT. 

IF YOU ARE UNSURE ABOUT WHAT YOU ARE DOING, DO NOT CONTINUE. SEEK QUALIFIED HELP IF 
NECESSARY. 

BEFORE RUNNING THE PROGRAMS DESCRIBED IN THIS DOCUMENT, BE SURE YOU HAVE ADEQUATE 
BACKUPS OF ALL IMPORTANT DATA THAT YOU ARE WORKING WITH. 

IMPORTANT 
 
 



4 
Run IBM i Access Client Solutions Data Transfers on the IBM i 

Copyright © 2012, Craig Pelkie, ALL RIGHTS RESERVED 

Why 

At each major section of this document, you will see a heading labeled Why. Within that heading, there is 
a brief description of why you need to perform the steps in the section. 

Although it looks like performing all of the steps in this document will be a laborious task, you should be 
able to complete all of the steps and have Data Transfer running on your IBM i within a few hours. By 
explaining why you need to perform the steps shown in a section, you should have a better understanding 
of what you are trying to accomplish within that section and how the steps fit in with the rest of the 
procedures described in this document. 

Submit a Data Transfer request to batch 

Why: You can run a Data Transfer request as a batch job. By moving the request to batch, you make it 
easier to create automated Data Transfer requests. 

Use the QSH command to invoke the Data Transfer 
Why: You can run the QSH command that invokes the java command to run the Data Transfer in batch. 

 _____ On a 5250 command line, enter the following SBMJOB command: 

SBMJOB CMD(QSH CMD('java -jar /IAC/acsbundle.jar   
                    -Dos400.class.path.security.check=0   
                    /PLUGIN=download   
                    /file=/IACXFER/xqcustcdt.dtfx')) + 
       JOB(QSHXFER) +  
       LOG(4 00 *SECLVL) +  
       LOGCLPGM(*YES) 

Note: the JOB, LOG and LOGCLPGM parameter values are optional, you do not need to enter values for those 
parameters. 

Use the RUNJVA command to invoke the Data Transfer 
Why: You can run the RUNJVA command to run the Data Transfer in batch.  

 _____ On a 5250 command line, enter the following SBMJOB command: 

SBMJOB CMD(RUNJVA CLASS('/IAC/acsbundle.jar')  
                  PARM('/PLUGIN=download'  
                       '/file=/IACXFER/xqcustcdt.dtfx')  
                  CHKPATH(*IGNORE)) +  
       JOB(RUNJVADTXF) +  
       LOG(4 00 *SECLVL) +  
       LOGCLPGM(*YES)  

Note: the JOB, LOG and LOGCLPGM parameter values are optional, you do not need to enter values for those 
parameters. 

The Command Line Download option 

Why: If you want to download all of the records and all fields in a database file to a stream file, you can 
use the Command Line Download plugin. You do not have to specify a .dtfx file (data transfer description 
file) with this plugin. 

In addition to the download and upload plugin types, IBM provides the cldownload plugin. The 
cldownload plugin provides a simplified way to download data from an IBM i database file to a stream file. 
Because the cldownload plugin downloads all of the rows and fields that are in the database file, there is 
no need to specify a data transfer description file (.dtfx). 

The syntax of the command is: 

java  –jar  acsbundle.jar   
      -Dos400.class.path.security.check=0   
       /PLUGIN=cldownload   
       /system=<system> 
       /hostfile=<library>/<file>(<member>) 
       /clientfile=<path><filename>.<extension> 

 

The /system parameter value is set to the name of the IBM i system where the database file is located. 
When you run the cldownload plugin on the IBM i, you can specify the system name LOCALHOST to 
download a database file from the system where the plugin is running. 

The /hostfile parameter is used to specify the qualified name (system format) of the database file to 
download from. If you need to specify a member name, enclose the name in parentheses. 

The /clientfile parameter specifies the complete path, file name and extension of the stream file that is 
created as a result of the download. The file extension is used to determine the type of transfer to 
perform. For example, if the file extension .xlsx is specified, an Excel 2007/2010 file is created. 

Create and work with the XFRDTA command 

Why: The Transfer Data (XFRDTA) command provides a simplified way to invoke the Data Transfer 
program. 

Review and compile the XFRDTAC CL program 

Why: The XFRDTAC CL program is the Command Processing Program for the XFRDTA command. 

The source code for this program can be downloaded at: 

http://www.web400.com/downloads/ACS/xfrdta.clp.txt 

Use the following CL command to compile this program: 

CRTCLPGM  PGM(your_lib/XFRDTAC)  SRCFILE(your_lib/QCLSRC) 

 
********************************************************************/ 
/* Program XFRDTAC                                                  */ 
/* Run the IBM i Access Client Solutions Data Transfer program      */ 
/********************************************************************/ 
/*                                                                  */ 
/* Copyright 2012, Craig Pelkie, ALL RIGHTS RESERVED                */ 
/* craig@web400.com                                                 */ 
/*                                                                  */ 
/* V1.01 - 2012-08-28 ORIGINAL VERSION                              */ 
/* V1.03 - 2012-11-14 UPDATE FOR OCT. 2012 ACSBUNDLE.JAR            */ 
/*                                                                  */ 
/********************************************************************/ 
/* You may freely make use of this program in any application that  */ 
/* you develop, as long as you retain the copyright notice shown    */ 
/* above.                                                           */ 
/*                                                                  */ 
/* This program is provided for instructional purposes only.        */ 
/* No warranty is expressed or implied.                             */ 
/*                                                                  */ 
/* Craig Pelkie explicitly disclaims all responsibilty for your     */ 
/* use of the program, and will not be held liable for any          */ 
/* results or unforeseen consequences that result from your         */ 
/* use of this program or any of the techniques that are            */ 
/* demonstrated by this program.                                    */ 
/*                                                                  */ 
/* By using this program, you acknowledge that:                     */ 
/*                                                                  */ 
/*   1. The techniques shown in this program have not been          */ 
/*      tested, documented, approved, or recommended by             */ 
/*      IBM. You should not expect any IBM support for              */ 
/*      the techniques shown in this program.                       */ 
/*                                                                  */ 
/*   2. Running this program may result in changes to files         */ 
/*      on your system. The expected outputs include the            */ 
/*      data transfer output file in your IBM i IFS, and            */ 
/*      possibly one or more job log files. It is your              */ 
/*      responsibility to verify the outputs of running the         */ 
/*      program, and to ascertain that no other unwanted or         */ 
/*      unexpected changes occur on your IBM i as a result          */ 
/*      of running this program.                                    */ 
/*                                                                  */ 
/*   3. No support from Craig Pelkie is expressed or implied        */ 
/*      in the provision of this program and its accompanying       */ 
/*      documentation, if any.                                      */ 
/*                                                                  */ 
/*   4. If you do not understand how to use this program and        */ 
/*      the potential consequences of running it, DO NOT RUN IT.    */ 
/********************************************************************/ 
 
 XFRDTAC:    PGM        PARM(&jardir + 
                             &plugin + 
                             &xfrdsc + 
                             &file   + 
                             &mbr    + 
                             &clientfile) 

(continues) 

             /*******************************************************/ 
             /* The path, file name and extension of the stream     */ 
             /* file to transfer data to.                           */ 
             /*                                                     */ 
             /* This is only used with &direction CLDOWNLOAD        */ 
             /* (Command Line Download)                             */ 
             /*                                                     */ 
             /* Supported extensions:                               */ 
             /*                                                     */ 
             /*     .csv  - Comma Separated Variable                */ 
             /*     .ods  - Open Document Spreadsheet               */ 
             /*     .xls  - Microsoft Excel 97 - 2003               */ 
             /*     .xlsx - Microsoft Excel 2007, 2010              */ 
             /*                                                     */ 
             /* If an unsupported extension is specified, the       */ 
             /* default extension .csv is used.                     */ 
             /*******************************************************/ 
             dcl        &clientFile  *char 256 
 
             /*******************************************************/ 
             /* The qualified name of the IBM i database file to    */ 
             /* transfer data from. This is only used with          */ 
             /* &plugin value *CLDOWNLOAD (Command Line Download)   */ 
             /*******************************************************/ 
             dcl        &file  *char 20 
 
             /*******************************************************/ 
             /* The IFS directory where the acsbundle.jar file      */ 
             /* is located                                          */ 
             /*******************************************************/ 
             dcl        &jardir *char 256 
 
             /*******************************************************/ 
             /* The member in the IBM i database file to            */ 
             /* transfer data from. This is only used with          */ 
             /* &plugin value *CLDOWNLOAD (Command Line Download)   */ 
             /*******************************************************/ 
             dcl        &mbr  *char 10 
 
             /*******************************************************/ 
             /* The plugin to use for the transfer.                 */ 
             /*                                                     */ 
             /*   *DOWNLOAD - transfer data from an IBM i database  */ 
             /*               file to a stream file                 */ 
             /*                                                     */ 
             /*   *CLDOWNLOAD - transfer the entire IBM i database */ 
             /*               file to a stream file                 */ 
             /*                                                     */ 
             /*   *UPLOAD   - transfer data from a stream file to   */ 
             /*               an IBM i database file                */ 
             /*******************************************************/ 
             dcl        &plugin *char 11 
 
             /*******************************************************/ 
             /* Variable used to set the value of the CLASS parm    */ 
             /* &jardir/acsbundle.jar                               */ 
             /*******************************************************/ 
             dcl        &valCLASS *char 270 
 
             /*******************************************************/ 
             /* Variable used to set the value of the /file parm    */ 
             /* /file=<filename>                                    */ 
             /*******************************************************/ 
             dcl        &valfile *char 262 
 
             /*******************************************************/ 
             /* Variable used to set the value of the client file.  */ 
             /* clientfile=<filename>                               */ 
             /*******************************************************/ 
             dcl        &valCfile *char 268 
 
             /*******************************************************/ 
             /* Variable used to set the value of the host file/lib */ 
             /* hostfile=<library>/<file>                           */ 
             /* hostifle=<library>/<file>(<member>)                 */ 
             /*******************************************************/ 
             dcl        &valHfile *char 43 

(continues) 

             /*******************************************************/ 
             /* Variable used to set the value of the /PLUGIN parm  */ 
             /*******************************************************/ 
             dcl        &valPLUGIN *char 18 
 
             /*******************************************************/ 
             /* Variable used to set the value of the /system parm  */ 
             /* Used only with cldownload plugin                    */ 
             /*******************************************************/ 
             dcl        &valSystem *char 17 + 
                        value('/system=LOCALHOST') 
 
             /*******************************************************/ 
             /* The path (directory) and file name of the transfer  */ 
             /* description file                                    */ 
             /*                                                     */ 
             /* Example: file for upload                            */ 
             /*          .dttx = "transfer to IBM i"                */ 
             /*                                                     */ 
             /*    /path/upload_description.dttx                    */ 
             /*                                                     */ 
             /* Example: file for download                          */ 
             /*          .dtfx = "transfer from IBM i"              */ 
             /*                                                     */ 
             /*    /path/download_description.dtfx                  */ 
             /*                                                     */ 
             /*******************************************************/ 
             dcl        &xfrdsc  *char 256 
 
             /*******************************************************/ 
             /* Set the parameter values to pass to the data        */ 
             /* transfer program                                    */ 
             /*******************************************************/ 
             chgvar     var(&valCLASS) + 
                        value(&jardir *tcat '/acsbundle.jar') 
 
             /*******************************************************/ 
             /* Set values for cldownload (Command Line Download)   */ 
             /*******************************************************/ 
             if         (&plugin *eq '*CLDOWNLOAD') then(do) 
 
                chgvar     var(&valPLUGIN) + 
                           value('/PLUGIN=cldownload') 
 
                chgvar     var(&valHfile) + 
                           value('/hostfile='      *tcat + 
                                 %sst(&file 11 10) *tcat + 
                                 '/'               *tcat + 
                                 %sst(&file  1 10)) 
 
                if  (&mbr *ne '*FIRST') then(do) 
                    chgvar  var(&valHfile) + 
                            value(&valHfile *tcat + 
                                  '('       *tcat + 
                                  &mbr      *tcat + 
                                  ')') 
                enddo 
 
                chgvar     var(&valCfile) + 
                           value('/clientfile=' *tcat + 
                                 &clientfile) 
 
             enddo 
 
             /*******************************************************/ 
             /* Set values for download / upload                    */ 
             /*******************************************************/ 
             if         (&plugin *eq '*DOWNLOAD') then(do) 
                chgvar     var(&valPLUGIN) + 
                           value('/PLUGIN=download') 
             enddo 
 
             if         (&plugin *eq '*UPLOAD') then(do) 
                chgvar     var(&valPLUGIN) + 
                           value('/PLUGIN=upload') 
             enddo 
 
             if         ((&plugin *eq '*DOWNLOAD') *or + 
                         (&plugin *eq '*UPLOAD  ')) then(do) 
 
                chgvar     var(&valfile) + 
                           value('/file=' *tcat &xfrdsc) 
             enddo 

(continues) 

             /*******************************************************/ 
             /* Run the data transfer                               */ 
             /*******************************************************/ 
             if (&plugin *eq '*CLDOWNLOAD') then(do) 
 
                runjva  class(&valCLASS) + 
                        parm(&valPLUGIN &valSystem &valHfile &valCfile) + 
                        chkpath(*IGNORE) 
                monmsg     msgid(CPF0000) 
             enddo 
 
             else cmd(do) 
 
                runjva  class(&valCLASS) + 
                        parm(&valPLUGIN &valfile) + 
                        chkpath(*IGNORE) 
                monmsg     msgid(CPF0000) 
             enddo 
 
             endpgm 

Review and create the XFRDTA command 

Why: The XFRDTA command can be used to run the Data Transfer program (XFRDTAC). 

The source code for this command can be downloaded at: 

http://www.web400.com/downloads/ACS/xfrdta.cmd.txt 

Use the following CL command to compile this program: 

CRTCMD  CMD(your_lib/XFRDTA)  PGM(your_lib/XFRDTAC)  SRCFILE(your_lib/QCMDSRC) 

/********************************************************************/ 
/* Command XFRDTA - transfer data using download/upload             */ 
/*                                                                  */ 
/* Copyright 2012, Craig Pelkie, ALL RIGHTS RESERVED                */ 
/********************************************************************/ 
 XFRDTA:     CMD        PROMPT('Transfer Data') 
 
             PARM       KWD(JARDIR) TYPE(*PNAME) LEN(256) MIN(1) + 
                          VARY(*NO) CASE(*MIXED) PROMPT('Directory + 
                          for acsbundle.jar') 
 
             PARM       KWD(PLUGIN) TYPE(*CHAR) LEN(11) + 
                          RSTD(*YES) VALUES(*DOWNLOAD *UPLOAD + 
                          *CLDOWNLOAD) MIN(1) PROMPT('Plugin  + 
                          to use') 
 
             PARM       KWD(XFRDSC) TYPE(*PNAME) LEN(256) + 
                          MIN(0) CASE(*MIXED) + 
                          PMTCTL(NOTCLD) PROMPT('Path to .dtfx / + 
                          .dttx file') 
 
             PARM       KWD(FILE) TYPE(QUAL1) MIN(0) PMTCTL(ISCLD) + 
                          PROMPT('Database file library to xfer') 
 
             PARM       KWD(MBR) TYPE(*NAME) LEN(10) DFT(*FIRST) + 
                          SPCVAL((*FIRST)) PMTCTL(ISCLD) PROMPT('Member') 
 
             PARM       KWD(CLIENTFILE) TYPE(*PNAME) LEN(256) + 
                          SPCVAL((*NONE)) MIN(0) + 
                          CASE(*MIXED) PMTCTL(ISCLD) + 
                          PROMPT('Client file path/name to write') 
 
 QUAL1:      QUAL       TYPE(*NAME) LEN(10) 
             QUAL       TYPE(*NAME) LEN(10) PROMPT('Database file name') 
 
 ISCLD:      PMTCTL     CTL(PLUGIN) COND((*EQ *CLDOWNLOAD)) 
 
 NOTCLD:     PMTCTL     CTL(PLUGIN) COND((*NE *CLDOWNLOAD))            

Parameters used with the XFRDTA command 

The parameters used with the Transfer Data command are described below. 

Parameter Value Usage 
JARDIR /path_name The path in the IFS where the 

acsbundle.jar file is located. Do not 
enter the name of the .jar file. 

PLUGIN *DOWNLOAD 

*UPLOAD 

*CLDOWNLOAD 

 

The type of transfer to perform. 

The following parameter is used when the value of the PLUGIN parameter is *DOWNLOAD or 
*UPLOAD. 

XFRDSC /path_name/file_name.dtfx 

/path_name/file_name.dttx 
The path in the IFS where the .dtfx 
(download) or .dttx (upload) 
description file is located. 

The following parameters are used when the value of the PLUGIN parameter is *CLDOWNLOAD. 

FILE Qualified name of an IBM i database 
file 

This is the name of the database file 
that is to be downloaded. The entire 
content of the file is download (all rows, 
all fields). 

MBR *FIRST 

member_name 

The name of the member in the file 
specified in the FILE parameter to 
download. 

CLIENTFILE /path_name/file_name.extension The complete path and file name of the 
file that is to be created as a result of 
the download. 

The extension must be a valid extension 
for the data transfer download (.xls, 
.xlsx, .csv, .ods, .txt). 

If an invalid extension is specified, the 
default extension value .csv is used. 

Work with the VNC Server 

Why: 

The IBM i Access Client Solutions product is designed to be used in a workstation environment, which 
provides a graphical user interface (GUI). When you run the ACS programs, the Java code in the product 
displays status messages and dialogs using Java GUI features. 

When you run the ACS Data Transfer program on the IBM i, there is no support for the Java GUI. If you 
simply try to run the Data Transfer program on the IBM i, it ends in error. 

In some cases, you can run Java code on the IBM i and indicate that you do not want to use the GUI. That 
is referred to as running in headless mode. You can find a description of that mode in the IBM Developer 
Kit for Java documentation at this link: 

http://publib.boulder.ibm.com/infocenter/iseries/v7r1m0/index.jsp?topic=%2Frzaha%2Fnawtsupport.htm 

It would be ideal if the ACS programs could use headless mode, as using it is as easy as adding a 
command line parameter to the java command (java.awt.headless=true). Unfortunately, even if you add 
that parameter value to the java command, the program ends in error. 

The alternative is to use what IBM calls the Native Abstract Windowing Toolkit (NAWT) that is also 
described in the IBM Developer Kit for Java documentation at this link: 

http://publib.boulder.ibm.com/infocenter/iseries/v7r1m0/index.jsp?topic=%2Frzaha%2Fguiintro.htm 

The ACS programs will run on your IBM i when you configure NAWT. The NAWT support is provided by 
using the Virtual Network Computing (VNC) server that is part of the IBM Tools for Developers LPP. The 
VNC server runs in the IBM i Portable Application Solutions Environment (PASE), which provides support 
for running AIX applications in the IBM i environment. 

After installing and configuring the VNC server on your IBM i, you will work with a VNC viewer on your 
workstation. When you use the ACS programs, the output of the programs is directed through the VNC 
server to the VNC viewer. 

Although it may sound very complicated to work with the tools, there are only a few steps that you need 
to do to get everything installed and configured. The steps are shown in the following sections. Once you 
have the VNC server and VNC viewer configured, you can use the ACS Data Transfer programs directly on 
your IBM i. 

What is the ps command and what do the gaxuw options mean? 

The ps command, used in the PASE environment, displays the status of processes. It is similar to the IBM i 
WRKACTJOB command. 

The gaxuw options that are used on the command are a set of five individual options. The options are 
described in the following IBM document, in the section titled Options: 

http://publib.boulder.ibm.com/infocenter/pseries/v5r3/index.jsp?topic=/com.ibm.aix.cmds/doc/aixcmds4/
ps.htm 

The options are used as follows. The descriptions of the options are from the IBM document linked to 
above: 

g — displays all processes 

a — displays information about all processes with terminals (ordinarily only the own processes of 
the user are displayed) 

x — displays processes without a controlling terminal in addition to processes with a controlling 
terminal 

u — displays user–oriented output. This includes the USER, PID, %CPY, %MEM, SZ, RSS, TTY, 
STAT, STIME, TIME and COMMAND fields. 

w — specifies a wide–column format for output (132 columns rather than 80).  

Remove environment variable(s) 
Why: If the environment variable(s) that are needed by the Data Transfer program are already in use, 
you can remove the variable(s) or change its value. This section shows how to remove the environment 
variable(s). 

 _____ On the Work with Environment Vars display (Figure 83), look for an environment variable 
named DISPLAY and an environment variable named XAUTHORITY. 

 _____ If the DISPLAY and XAUTHORITY environment variables are not defined, go to the section Add 
environment variables on page 78. 

 _____ Make a note of the current value of the DISPLAY or XAUTHORITY environment variables. 

 _____ You can either remove and add the environment variables, or you can change the value of the 
existing DISPLAY and XAUTHORITY environment variables. 

If you want to change the existing value of the existing environment variables, go to the section 
Change environment variable(s) on page 77. 

 _____ Use either of the following steps to remove the existing environment variables. 

Note: use the steps shown here to remove the DISPLAY and XAUTHORITY environment variables only. 
Do not remove any other environment variables that are shown in the WRKENVVAR display. 

 ____ Use option 4 (Remove) on the WRKENVVAR panel to remove the environment variable. 

 ____ Enter the command RMVENVVAR ENVVAR(DISPLAY) or RMVENVVAR ENVVAR(XAUTHORITY) 
on a 5250 command line. 

 _____ Continue with the steps in the section Add environment variables on page 78. 

  



77 
Run IBM i Access Client Solutions Data Transfers on the IBM i 

Copyright © 2012, Craig Pelkie, ALL RIGHTS RESERVED 

Change environment variable(s) 

Why: If the environment variables DISPLAY or XAUTHORITY are already defined, you can use the Change 
Environment Variable (CHGENVVAR) command to set the value that is required by the ACS Data Transfer 
program. 

 _____ Use either of the following steps to change the existing environment variables DISPLAY and 
XAUTHORITY. 

 ____ Use option 2 (Change) on the WRKENVVAR panel to change the value. 

 ____ Use the Change Environment Variable (CHGENVVAR) command on a 5250 command line. 

 _____ Set the values of the environment variables as follows. 

Environment Variable Value 

DISPLAY ibm_i_host_name:vnc_instance_number 

or 

ibm_i_tcp_ip_address:vnc_instance_number 

Where vnc_instance_number is the instance number 
assigned when you started the VNC server (see Figure 
72 on page 66 for an example of determining the 
instance number). 

XAUTHORITY 

 
/home/your_user_profile_name/.Xauthority 
 

For your_user_profile_name, enter the name of the 
user profile where the .Xauthority file is located (see 
Figure 74 on page 68). 

Add environment variables 

Why: The ACS Data Transfer program uses two environment variables named DISPLAY and XAUTHORITY. 
The values of the environment variable point to the VNC server and its configuration. 

 _____ Enter the following commands on a 5250 command line: 

ADDENVVAR ENVVAR(DISPLAY) VALUE(ꞌibm_i_host_name:vnc_instance_numberꞌ) 

or 

ADDENVVAR ENVVAR(DISPLAY) VALUE(ꞌibm_i_tcp_ip_address:vnc_instance_numberꞌ) 

 

ADDENVVAR ENVVAR(XAUTHORITY) VALUE(ꞌ/home/user_profile_name/.Xauthorityꞌ) 

Notes 

 Enter the environment variable names in upper–case, as shown in the commands above. 

 The vnc_instance_number is the instance number assigned when you started the VNC server (see 
Figure 72 on page 66 for an example of determining the instance number). 

 For your_user_profile_name, enter the name of the user profile where the .Xauthority file is located 
(see Figure 74 on page 68). 

 

 
IMPORTANT! 

The values that you enter for the environment variables in your 5250 session are not 
persistent. When you sign-off from your 5250 session, the values for the environment variables 
are not kept. 

The next time you sign on to a 5250 session, you need to add the environment variables. 

If you will be running Data Transfers frequently from your 5250 session, you may want to 
consider adding the ADDENVVAR commands to your start up program. 

IMPORTANT! 

Start the Data Transfer program 
Why: You can now run the Data Transfer program. For the first test, you will run the java command in the 
QSHELL environment. 

 _____ Verify that the VNC viewer application is still running on your workstation. If it is not running, start 
it and connect to the VNC server running on your IBM i. (See the section Work with the VNC 
Viewer on page 72.) 

 _____ On a 5250 command line, enter the QSH (Start QSHELL) command. 

 _____ The QSH Command Entry panel is displayed. 

 _____ Enter the following command on the QSH Command Entry line. 

Note: the command is case–sensitive, enter it as shown. 

Note: the value for the os400.class.path.security.check parameter is zero, not the letter “O”. 

Note: for the /userid parameter, enter the name of your user profile. This must be the same user 
profile name that you used to start the VNC server. 

Note: enter the entire command on one line in the QSH Command Entry panel. The line will “wrap” 
on the Command Entry panel as you type it in. The command text shown in this example is 
wrapped on this page. 

java  –jar  /IAC/acsbundle.jar  -Dos400.class.path.security.check=0   
 
   /PLUGIN=download  /file=/IACXFER/xqcustcdt.dtfx   
 
   /userid=your_user_profile_name 

